

UTILISATION

Création d'une nouvelle expérience

Dans cet environnement de simulation, plusieurs expériences sont déjà installées pour permettre à l'utilisateur d'effectuer plusieurs types de tests rapidement. Les expériences disponibles dans l'environnement sont l'expérience pour tester les insertions de voie, l'expérience pour tester le freinage en cas de collision proche entre deux véhicules, l'expérience pour tester les ronds-points, et enfin une expérience un peu plus globale contenant un circuit avec plusieurs utilisations de règles du code de la route (insertions, croisement etc...).

Pour créer une nouvelle expérience, plusieurs étapes doivent être réalisées :

- Créer le schéma des routes avec un fichier shapefile depuis un éditeur spécialisé (exemple : QGIS)
- Créer les locations des différents checkpoints avec un fichier shapefile depuis un éditeur spécialisé (exemple : QGIS)
- Dans le fichier Main.gaml de la simulation :
 - o Ajouter dans la fonction string `getRoadPathFile()` le chemin du fichier shapefile représentant les routes (ajout d'un match dans le switch sur `expChoice`)
 - o Ajouter dans la fonction string `getCheckpointPathFile()` le chemin du fichier shapefile représentant les checkpoints (ajout d'un match dans le switch sur `expChoice`)
 - o Dans la partie « ACTIONS » du code, ajouter une action qui permet de créer les agents pour l'expérience en question grâce aux différents constructeurs disponibles
 - o Dans le bloc d'instruction « init » encore une fois dans le switch sur `expChoice`, ajouter un appel de l'action développée sur le dernier point avec un nouveau match avec le même numéro utilisé dans les fonctions `getRoadPathFile` et `getCheckpointPathFile` (voir captures d'écran ci-dessous)
- Dans le fichier `trafficModel.experiment`, augmenter d'un la valeur maximale du paramètre de l'expérience « Number of the experiment » associé à la variable `expChoice`

```

experiment trafficModel model: 'Main.gaml' type:gui {
    parameter "Cycle time to change checkpoint location (hours)" var: cycle_time_checkpoint min: 1 max: 5 category: "Checkpoints";
    parameter "Number of the experiment" var: expChoice min: 0 max: 4 category: "Experiment";
    . . .
}

```

Dans notre plateforme d'expérimentation, plusieurs expériences sont déjà disponibles et le choix s'effectue grâce à la variable `expChoice` présente dans le fichier `Main.gaml`, les différentes valeurs possibles pour `expChoice` sont :

- 0 : `basicExp`, une expérience basique avec quelques éléments tels que croisements, insertions
- 1 : `brakelfCollisionExp` : expérience pour tester que lorsque deux véhicules vont rentrer en collision, ils freinent tous les deux pour éviter de se rentrer dedans
- 2 : `insertionExp` : expérience pour tester l'insertion dans une voie
- 3 : `rondPointExp` : expérience pour tester les comportements des véhicules lors de l'utilisation d'un rond-point
- 4 : `UPSCampusExp` : expérience sur une carte plus grande (carte du campus de l'université Paul Sabatier)

Captures d'écran d'un exemple de création d'expérience (Ajout de l'expérience pour le test d'un rond-point)

```

/* Retourne le chemin du fichier shapefile des routes en fonction de l'experience choisie
 * En cas d'ajout d'experience: ajouter ici un nouveau chemin
 */
string getRoadPathFile {
    switch expChoice {
        match 0 { return "../includes/shapefiles/circuitv2.shp"; }
        match 1 { return "../includes/shapefiles/TestBrakeIfCollision.shp"; }
        match 2 { return "../includes/shapefiles/TestInsertion.shp"; }
        match 3 { return "../includes/shapefiles/TestRondsPoints.shp"; }
        default { return "erreurExpChoice"; }
    }
}

/* Retourne le chemin du fichier shapefile des checkpoints en fonction de l'experience choisie
 * En cas d'ajout d'experience: ajouter ici un nouveau chemin
 */
string getCheckpointPathFile {
    switch expChoice {
        match 0 { return "../includes/shapefiles/checkpointsv2.shp"; }
        match 1 { return "../includes/shapefiles/TestBrakeIfCollisionCheckpoints.shp"; }
        match 2 { return "../includes/shapefiles/TestInsertionCheckpoints.shp"; }
        match 3 { return "../includes/shapefiles/TestRondsPointsCheckpoints.shp"; }
        default { return "erreurExpChoice"; }
    }
}

```

Ajout dans les fonctions `getRoadPathFile` et `getCheckpointPathFile` des chemins des nouveaux shapefiles

```

/* Creation des agents pour l'experience de test des ronds points
 * Agents: 3 agents qui vont devoir se servir du rond point pour
 * atteindre leur checkpoint cible
 */
action initRondPointExp {
  do consBusWithStartAndTarget(0.2, Checkpoint.population[0].location, Checkpoint.population[3]);
  do consBusWithStartAndTarget(0.2, Checkpoint.population[1].location, Checkpoint.population[4]);
  do consBusWithStartAndTarget(0.2, Checkpoint.population[2].location, Checkpoint.population[4]);
}

```

Création de l'action initRondPointExp qui crée les agents pour cette expérience

```

/* Algorithme d'initialisation:
 * - creer les routes et checkpoints en fonction de l'experience
 *   choisie
 * - initialiser le graphe des routes
 * - creer les agents correspondant a l'experience choisie
 */
init {
  create Road from: roads_shapefile;
  create Checkpoint from: checkpoints_shapefile;
  roadGraph <- as_edge_graph(Road.population);
  switch expChoice{
    match 0 { do initBasicExp; break; }
    match 1 { do initBrakeIfCollisionExp; break;}
    match 2 { do initInsertionExp; break; }
    match 3 { do initRondPointExp; break; }
    default { write "erreur expChoice"; break; }
  }
}

```

Ajout de l'appel de l'action codée juste au-dessus